

Wikidata Query Service/User Manual/MWAPI

[< Wikidata Query Service](#) | [User Manual](#)

Contents

Mediawiki API Query Service

- Service

- Input

- Pagination

- Output

- Supported services

Configuration

- Individual service configuration

- Input

- Output

Examples

- Find all entities with labels "cheese" and get their types

- Find articles in Wikipedia

- Find articles in Wikipedia speaking about cheese and see which Wikibase items they correspond to

- Find category members

Mediawiki API Query Service

Mediawiki API Service allows to call out to [Mediawiki API](#) from SPARQL, and receive the results from inside the SPARQL query.

Service

The query is initiated by SERVICE with URL `wikibase:mwapi`.

Input

Query inputs are specified as service parameters, in clause looking like:

```
bd:serviceParam mwapi:language "en" .
```

This defines parameter "language" in the query.

There are two special parameters, `api` and `endpoint`, which are always required and identify the endpoint host (wiki) and the specific API used. The API endpoints are pre-defined in configuration, see below. Currently, only pre-configured endpoints are allowed.

Example:

```
bd:serviceParam wikibase:api "EntitySearch" .  
bd:serviceParam wikibase:endpoint "www.wikidata.org" .
```

The rest of the parameters use predicate starting with `mwapi:` and the name is the name of the parameter in the query, e.g. `mwapi:srsearch`. The value is the value that will be used in the query, it can be a constant or a variable. If the variable is not bound when the query is run, the parameter is skipped or, if it is required, as defined by the configuration, an error happens.

It is permissible to add input parameters not specified in the configuration, they will be passed to the service query. Please refer to the [API documentation](#) for the lists of parameters each service has. The configuration also defines some default parameters – if you override them in the service parameters, you may want to merge the default value with your desired value to ensure the output looks as expected by MWAPI (e.g. use `mwapi:ppprop "wikibase_item|page_image_free"` instead of just `mwapi:ppprop "page_image_free"`).

Currently supported endpoints are: `*.wikipedia.org`, `commons.wikimedia.org`, `www.mediawiki.org`, `www.wikidata.org`, `test.wikidata.org`. If any other wikis need to be supported, please [leave a comment to the developers](#) and we will enable them.

Pagination

By default, WDQS will try to fetch all the results of the query, with continuations. You can set the page size by using service-specific limit parameters, but the service will use continuation to fetch the following pages. This can be controlled by the `wikibase:limit` parameter, which will limit total number of rows fetched in one API request:

```
bd:serviceParam wikibase:limit 50 .
```

This will fetch only 50 rows, regardless of pagination settings. Note that SPARQL `LIMIT` clause will cause the same effect, but will be applied only after the API call has finished - all the data has been fetched and then the extra data has been discarded. It is more efficient to use the internal limit.

A special value of "once" disables the continuation mechanism:

```
bd:serviceParam wikibase:limit "once" .
```

This makes the service to stop after the first API call and not use continuations. The number of results depends on service parameters.

There is a hard-coded limit of 10000 rows which always applies.

Output

Mediawiki API services are asked to return the results in XML, from which they are extracted to output parameters using XPath. There are two ways to specify output parameters: by referring to the configured parameter and by specifying XPath directly.

The configured output parameter would be used like this:

```
?title wikibase:apiOutput mwapi:title .
```

This results in variable `?title` bound to the XPath result defined by `"title"` output variable. Note that the configured output parameters expect certain input parameters to be present, and if you override the configured defaults the output parameters may break: for example, the configuration specifies a default parameter of `prop=info|pageprops`, and if you override it with `mwapi:prop "revisions"`, then the `mwapi:lastrevid` output parameter will no longer be able to get the `lastrevid` from the response. (You would fix this by looking up the default `prop` in the configuration and then changing your query to use `mwapi:prop "info|pageprops|revisions"`.)

The direct XPath expression is used like this:

```
?ns wikibase:apiOutput "@ns" .
```

In this case, the object of the triple is an XPath string and not `mwapi:` URI.

In both cases, XPath is evaluated relative to item path, specified in service configuration. There's no difference in XPath evaluation for both forms.

The predicates that can be used are `wikibase:apiOutput`, `wikibase:apiOutputItem` and `wikibase:apiOutputURI`. The first form results in a string taken literally; the second interprets the string as Item ID and constructs a full item URI; and the third interprets the string as an URI and constructs an URI value.

Predicate `wikibase:apiOrdinal` allows to define variable that would be set to the number of the result in the original API output. Note that due to the multithreaded nature of the SPARQL engine the results are not necessarily returned in the order in which they were received by the API, which may be useful for queries like search where result order is important. Using `wikibase:apiOrdinal` will allow to recover the original order. The object of the triple is currently ignored.

```
?num wikibase:apiOrdinal true.
```

Supported services

Currently the following services are supported:

Service	Inputs	Outputs	Description
<u>Generator</u>	generator , prop, pprop	title, item, pageid, lastrevid, timestamp	Call any generator API. Use "generator" parameter to specify, and specific generator parameters to further amend the search (see the example below).
<u>Categories</u>	titles , cllimit	category, title	Get a list of categories on the page.
<u>Search</u>	srsearch , srwhat, srlimit	title	Full-text search in wiki.
<u>EntitySearch</u>	search , language, type, limit	item, label	Wikibase entity search, by title.

Required parameters are in bold. Please refer to the service documentation (linked in Service column) for the meaning of input parameters.

The full list of services can be seen in the production config file (<https://github.com/wikimedia/wikidata-query-deploy/blob/master/mwservices.json>).

Configuration

The service is configured by a JSON file listing particular API configurations and a list of endpoints. Please see the example configuration.

On the top level, there is a list of endpoint hostnames under "endpoints" and API configs under "services", with the keys being service names for use with `wikibase:api`.

Only endpoints listed in `endpoints` are allowed. The list can also contain prefixes - i.e. `.wikipedia.org` allows any hostname that ends with `.wikipedia.org`.

Individual service configuration

The service configuration has two required elements - `params` for input parameters and `output` for output parameters. All the rest are currently ignored by the code, but `@note` and `@docs` are used to store description and link to documentation about particular API. It is safe to assume that config entries starting with `@` will be always ignored by the code.

Input

Input parameters are keyed by parameter name, and the value can be either constant - in which case the parameter is fixed, always present and its value can not be changed, or object, which contains description of variable parameter.

Variable parameter is bound by the query template or variable. The configuration specifies its type, which is currently ignored but may be enforced later. The configuration can also specify the `default`, which will be used in case the parameter is not specified or not bound. If the default is `" "`, the parameter will be omitted in case it is not specified or not bound. If no default is specified, the parameter is treated as required and omitting it will result in an error.

Output

Output is defined by `items`, which specifies the XPath for extracting specific result elements for the query, and set of output variables under `vars`. Each key in `vars` is variable name (to be used with `wikibase:apiOutput`) and the value is XPath to this particular value, calculated relative to the `items` path. Note that it is possible to also specify absolute paths for specific variables, but in that case you will get copies of that value for each result set.

Examples

Find all entities with labels "cheese" and get their types

- Properties used: subclass of (P279) (<https://www.wikidata.org/wiki/Property:P279>), instance of (P31) (<https://www.wikidata.org/wiki/Property:P31>)

```
SELECT * WHERE {
  SERVICE wikibase:mwapi {
    bd:serviceParam wikibase:api "EntitySearch" .
    bd:serviceParam wikibase:endpoint "www.wikidata.org" .
    bd:serviceParam mwapi:search "cheese" .
    bd:serviceParam mwapi:language "en" .
    ?item wikibase:apiOutputItem mwapi:item .
    ?num wikibase:apiOrdinal true .
  }
  ?item (wdt:P279|wdt:P31) ?type
} ORDER BY ASC(?num) LIMIT 20
```

Try it! (<https://query.wikidata.org/#SELECT%20%2A%20WHERE%20%7B%0A%20%20SERVICE%20wikibase%3Amwapi%20%7B%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aapi%20%22EntitySearch%22%20.%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aendpoint%20%22www.wikidata.org%22%20.%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Asearch%20%22cheese%22%20.%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Alanguage%20%22en%22%20.%0A%20%20%20%20%20%20%20%20%20%3Fitem%20wikibase%3AapiOutputItem%20mwapi%3Aitem%20.%0A%20%20%20%20%20%20%20%20%20%3Fnum%20wikibase%3AapiOrdinal%20true%20.%0A%20%20%20%7D%0A%20%20%20%3Fitem%20%28wdt%3AP279%7Cwdt%3AP31%29%20%3Ftype%0A%7D%20ORDER%20BY%20ASC%28%3Fnum%29%20LIMIT%2020>)

Find articles in Wikipedia

```
SELECT * WHERE {
  SERVICE wikibase:mwapi {
    bd:serviceParam wikibase:api "Search" .
    bd:serviceParam wikibase:endpoint "en.wikipedia.org" .
    bd:serviceParam mwapi:srsearch "cheese" .
    ?title wikibase:apiOutput mwapi:title .
  }
} LIMIT 20
```

Try it! (<https://query.wikidata.org/#SELECT%20%2A%20WHERE%20%7B%0A%20%20SERVICE%20wikibase%3Amwapi%20%7B%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aapi%20%22Search%22%20.%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aendpoint%20%22en.wikipedia.org%22%20.%0A%20%20%20%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Asrsearch%20%22cheese%22%20.%0A%20%20%20%20%20%20%20%20%20%3Ftitle%20wikibase%3AapiOutput%20mwapi%3Atitle%20.%0A%20%20%20%7D%0A%7D%20LIMIT%2020>)

Find articles in Wikipedia speaking about cheese and see which Wikibase items they correspond to

- Properties used: instance of (P31) (<https://www.wikidata.org/wiki/Property:P31>), subclass of (P279) (<https://www.wikidata.org/wiki/Property:P279>)

```
SELECT ?item ?type ?typeLabel ?typeLabel WHERE {
{
  SELECT ?item WHERE {
    SERVICE wikibase:mwapi {
      bd:serviceParam wikibase:endpoint "en.wikipedia.org" .
      bd:serviceParam wikibase:api "Generator" .
      bd:serviceParam mwapi:generator "search" .
      bd:serviceParam mwapi:gsrsearch "cheese" .
      bd:serviceParam mwapi:gsrlimit "max" .
      ?item wikibase:apiOutputItem mwapi:item .
    }
  } LIMIT 100
}
hint:Prior hint:runFirst "true".
?item wdt:P31|wdt:P279 ?type .
SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }
} LIMIT 100
```

Try it! (<https://query.wikidata.org/#SELECT%20%3Fitem%20%3Ftype%20%3FitemLabel%20%3FtypeLabel%20WHERE%20%7B%0A%20%7B%0A%20%20%20SELECT%20%3Fitem%20WHERE%20%7B%0A%20%20%20%20SERVICE%20wikibase%3Amwapi%20%7B%0A%20%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aendpoint%20%22en.wikipedia.org%22%20.%0A%20%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aapi%20%22Generator%22%20.%0A%20%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Agenerator%20%22search%22%20.%0A%20%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Agsrsearch%20%22cheese%22%20.%0A%20%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Agsrlimit%20%22max%22%20.%0A%20%20%20%20%20%20%20%3Fitem%20wikibase%3AapiOutputItem%20mwapi%3Aitem%20.%0A%20%20%20%20%20%7D%0A%20%20%7D%20LIMIT%20100%0A%20%7D%0A%20hint%3APrior%20hint%3ArunFirst%20%22true%22.%0A%20%3Fitem%20wdt%3AP31%7Cwdt%3AP279%20%3Ftype%20.%0A%20SERVICE%20wikibase%3Alabel%20%7B%20bd%3AserviceParam%20wikibase%3Alanguage%20%22en%22.%20%7D%0A%7D%20LIMIT%20100>)

Find category members

- Items used: parking lot (Q6501349) (<https://www.wikidata.org/wiki/Q6501349>)
- Properties used: topic's main category (P910) (<https://www.wikidata.org/wiki/Property:P910>)

```
SELECT * WHERE {
  wd:Q6501349 wdt:P910 ?category .
  ?link schema:about ?category; schema:isPartOf <https://en.wikipedia.org/>; schema:name ?title .
  SERVICE wikibase:mwapi {
    # in
    bd:serviceParam wikibase:api "Generator" .
    bd:serviceParam wikibase:endpoint "en.wikipedia.org" .
    bd:serviceParam mwapi:gcmtitle ?title .
    bd:serviceParam mwapi:generator "categorymembers" .
    bd:serviceParam mwapi:gcmprop "ids|title|type" .
    bd:serviceParam mwapi:gclimit "max" .
    # out
    ?member wikibase:apiOutput mwapi:title .
    ?ns wikibase:apiOutput "@ns" .
    ?item wikibase:apiOutputItem mwapi:item .
  }
}
```

```
# FILTER (?ns = "14")
}
```

Try it! (<https://query.wikidata.org/#SELECT%20%2A%20WHERE%20%7B%0A%20%20wd%3AQ6501349%20wdt%3AP910%20%3Fcategory%20.%0A%20%20%3Flink%20schema%3Aabout%20%3Fcategory%3B%20schema%3AisPartOf%20%3Chttps%3A%2F%2Fen.wikipedia.org%2F%3E%3B%20schema%3Aname%20%3Ftitle%20.%0A%20%20SERVICE%20wikibase%3Amwapi%20%7B%0A%20%20%20%20%20%23%20in%0A%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aapi%20%22Generator%22%20.%0A%20%20%20%20%20%20bd%3AserviceParam%20wikibase%3Aendpoint%20%22en.wikipedia.org%22%20.%0A%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Agcmtitle%20%3Ftitle%20.%0A%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Agcmprop%20%22ids%7Ctitle%7Ctype%22%20.%0A%20%20%20%20%20%20bd%3AserviceParam%20mwapi%3Agcmlimit%20%22max%22%20.%0A%20%20%20%20%20%20%23%20out%0A%20%20%20%20%20%20%3Fmember%20wikibase%3AapiOutput%20mwapi%3Atitle%20%20.%0A%20%20%20%20%20%20%3Fns%20wikibase%3AapiOutput%20%22%40ns%22%20.%0A%20%20%20%20%20%20%3Fitem%20wikibase%3AapiOutputItem%20mwapi%3Aitem%20.%0A%20%20%20%7D%0A%23%20FILTER%20%28%3Fns%20%3D%20%2214%22%29%0A%7D>)

Retrieved from "https://www.mediawiki.org/w/index.php?title=Wikidata_Query_Service/User_Manual/MWAPI&oldid=3098893"

This page was last edited on 17 February 2019, at 11:24.

Text is available under the [Creative Commons Attribution-ShareAlike License](#); additional terms may apply. See [Terms of Use](#) for details.